
14 Vector Calculus

14.1 Parametric Equations of Curves

In this section we will see how Maple V can be used to plot interesting curves using parametric representation.
First of all consider the circle centered at the origin with radius 2. The equation for this curve is

x2 + y2 = 4

If you wish plot this curve then you soon become aware that the set of points represented by the above equation
cannot satisfy a relationship of the form y = f (x) where f is a single valued function (the vertical line test fails).
The graph of the function

f1( x ) =
√

4 − x2

is the upper semi-circle of radius 2.

> P1 := plot(sqrt(4-x^2),x=-2..2):";

Whereas the following plot gives the bottom half:

> P2 := plot(-sqrt(4-x^2),x=-2..2):";

We can now use display to put both plots together. See Figure 70.

> with(plots):
> display({P1,P2},scaling =constrained);

We can obtain a parameterization for the entire circle. Since the following identity is true for all real values of t

-2

-1

0

1

2

-2 -1 0 1 2
x

Figure 70: Circle of Radius 2

Figure 71: Parabola x = t, y = t3 − 2t

sin( t )2 + cos( t )2 = 1

it follows that x = 2 sin(t), y = 2 cos(t) satisfy the equation

x2 + y2 = 4

for all values of t. Moreover, the image of the mapping defined by the parametric equations x = 2 cos(t), y =
2 sin(t) generate the entire circle, which is shown in Figure 70, as t varies for 0 to 2π with one plot statement using
the parametric plot syntax.

> plot([2*cos(t),2*sin(t),t=0..2*Pi],scaling=constrained);

If a curve is given as an explicit function of x or y then it is easy to write a parameterization: if a curve is
described by the relation y = f (x), for a ≤ x ≤ b, then a parameterization for this curve is x = t, y = f (t),
for a ≤ t ≤ b, and, similarly, if a curveis described as x = g(y), then the parameterization can be of the form
x = g(t), y = t. For example, suppose

y = x3 − 2 x

is given. Then either of the following plot commands will give the Figure 71. for any range, say, −3 ≤ x ≤ 3.

> plot(x^3-2*x,x=-3..3);
> plot([t,t^3-2*t,t=-3..3]);

Similarly if x = y2 − y is the function you wish to plot then the following command does the trick. See Fig-
ure 72.

> plot([t^2-t,t,t=-2..2]);

Figure 72: Parabola x = t2 − t, y = t

Figure 73: Line x = 1 + 2t, y = 3 + 5t

Straight lines are very easy to parameterize. For example if y = mx + b is the equation of the line then x =
t, y = mt + b gives a parameterization. But you can sometimes write the parametric equations directly from their
geometric description. For example, a portion of the straight line through the point (1,3) parallel to the vector <

2, 5 > has parametric equations x = 1 + 2t, y = 3 + 5t, is shown in Figure 73 and can be plotted by the command:

> plot([1+2*t,3+5*t,t=-1..1]);

On occasion it possible to plot curves that would be very complicated to express in even implicit algebraic terms.
For example, the following curve is known as a Lissajous Figure and is shown in Figure 74.

> plot([cos(3*t),sin(5*t),t=0..2*Pi],scaling=constrained);

Solutions of differential equations are usually represented in parametric form. For example x = cos(2t), y =
−2 sin(2t) is a solution of the initial value problem:

d
dt

x( t ) = x( t )

d
dt

y( t ) = −4 x( t )

x( 0 ) = 1, y( 0 ) = 0

as can easily be verified. A plot of this curve is given by

> plot([cos(2*t),-2*sin(2*t),t=0..Pi],scaling = constrained);

Figure 74: Lissajous Figure x = cos(3t), y = sin(5t)

Figure 75: Ellipse x= cos(2t), y= -2sin(2t)

See Figure 75. When plotting curves in three dimensions using Maple V we need to use the command spacecurve
which is part of the plots package. Consider the curve written parametrically as

x = cos t, y = sin t, z = t.

It’s plot, Figure 76 over the range [0, 2π] is given by

> spacecurve([cos(t),sin(t),t],t=0..2*Pi);

Figure 76: Cylindrical Helix: x = cos(t), y = sin(t), z=
t

Figure 77: A Closed Curve in Three Space

The complicated curve shown in Figure 77, is actually a simple closed curve, and can be obtained as follows:

> spacecurve([(4+sin(3*t))*cos(2*t),(4+sin(3*t))*sin(2*t),cos(3*t)],
> t=0..2*Pi);

Exercises 14.1

1. Make a Maple V plot of the conical helix given by the parametric equations

x = t cos(3t), y = t sin(3t), z = t, −∞ < t < ∞.

2. Write parametric equations for first octant portion of the curve of intersection of the sphere and cylinder

x2 + y2 + z2 = 64, x2 + (y − 4)2 = 16.

Make a Maple V plot of this curve.

3. Show that x = sin(2t) cos(t), y = sin(2t) sin(t), 0 ≤ t ≤ 2π is a parametric representation for the curve
given implicitly by the equation

(x2 + y2)3 = 4x2 y2.

Plot the graph of this curve using Maple V using the parametric representation and also using the explicit
repesentation. Is it worthwhile to obtain the parametric representation?

14.2 Parametric Equations of Surfaces

As with the implicit representation of a circle, the implicit representation of a sphere may require a piece meal
approach when making plots. In order to plot the graph of the sphere

x2 + y2 + z2 = 4

we could solve for z and obtain a parameterization for the lower half with

z = f1(x, y) = −
√

4 − x2 − y2, 0 ≤ x2 + y2 ≤ 4

and obtain a plot with plot3d.

> P1 := plot3d(-sqrt(4-x^2-y^2),x=-2..2,y=-sqrt(4-x^2)..sqrt(4-x^2),
> style=patchnogrid,numpoints=2500):";

In order to get the top half we type in the following:

> P2 := plot3d(sqrt(4-x^2-y^2),x=-2..2,y=-sqrt(4-x^2)..sqrt(4-x^2),
> style=patchnogrid,numpoints=2500):";

Finally, to put both plots, as shown in Figure 78, at the same time we write.

> with(plots):
> display({P1,P2},axes=normal,scaling=constrained);

> with(plots):
> display({P1,P2},scaling =constrained);

Figure 78: Sphere of Radius 2 Figure 79: Cone Plot with implicitplot3d

This is analogous to the procedure that was use in plotting a circle when its implicit equation is given in rect-
angular coordinates. Just as with the situation with a circle, there is a parameterization which enables us to plot
the graph with just one set of defining equations. From a well known trigonometric identity you can see that

x = 2 sin( u ) cos( v ), y = 2 sin( u ) sin( v ), z = 2 cos( u )

provides a parametric representation for the entire sphere if 0 ≤ u ≤ π, 0 ≤ v ≤ 2π. The sphere can thus be
plotted, as is also shown in Figure 79, as follows:

> plot3d([2*sin(u)*cos(v),2*sin(u)*sin(v),2*cos(u)],u=0..Pi,v=Pi..2*Pi,
> scaling=constrained,axes=normal,style=patch);

Example: Find a parameterizations for the bottom half of the cone that intersects the z = 0 plane in the circle

x2 + y2 = 4

and whose vertex is located a height 8 above the plane z=0.

Solution: An implicit equation for the entire cone is seen to be

4

(
1 − 1

8
z

)2

= x2 + y2

and thus the following use of implicitplot3d produces the plot, shown in Figure 79, of the surface.

> implicitplot3d(4*(1-z/8)^2=x^2+y^2,x=-3..3,y=-3..3,z=-1..17);

You can verify that if

x =
(

1 − 1
8

u

)
cos( v ), y =

(
1 − 1

8
u

)
sin( v ), z = u

then the point (x, y, z) lies on the cone. Thus the following command plots the bottom half of the cone.

> plot3d([(1-u/8)*cos(v),(1-u/8)*sin(v),u],u=0..8,v=0..2*Pi);

Figure 80: Cone Using Parametric Equations Figure 81: Plot of Torus

Example: Plot the torus which is created by revolving the circle

( x − 1 )2 + z2 = 4

about the z−axis. After verifying that the following gives a parameterization for this torus

x = 2 cos( u ) + cos( v ) cos( u ), y = 2 sin( u ) + cos( v ) sin( u ), z = sin( v )

it is easy to see that the the next command plots it. See Figure 81.

> plot3d([2*cos(u)+cos(v)*cos(u),2*sin(u)+cos(v)*sin(u),sin(v)],
> u=0..2*Pi,v=0..2*Pi);

Exercises 14.2

1. Find an implicit equation in rectangular coordinates for the surface given by

x = 2u cos v, y = 3u sin v, z = u.

Make a Maple V plot of the surface using the given parametric representation.

2. Write parametric equations for the hyperboloid of one sheet

x2 + y2 = z2 + 1

and make a Maple V plot that illustrates the qualitative features of this surface.

14.3 Velocity and Acceleration Vectors in Space

Consider a curve parameterized by t ∈ [a, b] :

x = f (t), y = g(t), z = h(t).

> with(linalg):

Warning: new definition for norm
Warning: new definition for trace

The position vector R for this function is given by

> R := vector([f(t),g(t),h(t)]);

R := [ f( t )g( t )h( t ) ]

The velocity vector V is obtained by differentiation, which since R is a vector requires the map command.

> V := map(diff,R,t);

V :=
[

∂

∂t
f( t )

∂

∂t
g( t )

∂

∂t
h( t )

]
The magnitude of the the velocity vector is the speed ds

dt .

> dsdt := sqrt(dotprod(V,V));

dsdt :=
((

∂

∂t
f( t )

)2

+
(

∂

∂t
g( t )

)2

+
(

∂

∂t
h( t )

)2
)1/2

The arc length of the curve is found through integrating the speed which gives the formula:

> Length := Int(s,t=a..b);

Length :=
∫ b

a

((
∂

∂t
f( t )

)2

+
(

∂

∂t
g( t )

)2

+
(

∂

∂t
h( t )

)2
)1/2

dt

As an example consider the curve parameterized by

x = (4 + sin t) cos 3t, y = (4 + sin t), z = sin 3t, cos(t)

The position vector R is

> R := vector([(4+sin(t))*cos(3*t),(4+sin(t))*sin(3*t),cos(t)]);

R := [ ( 4 + sin( t ) ) cos( 3 t ) ( 4 + sin( t ) ) sin( 3 t ) cos( t ) ]

A plot of this curve from t = 0 to t = π is shown in Figure 82 and is produced by

> with(plots):
> spacecurve(R,t=0..2*Pi);

Thus the tangent vector is:

> V := map(diff,R,t);

V := [cos( t ) cos( 3 t ) − 3 ( 4 + sin( t ) ) sin( 3 t )

cos( t ) sin( 3 t ) + 3 ( 4 + sin( t ) ) cos( 3 t ) − sin( t )]

The square of the speed is

> dsdt2 := dotprod(V,V);

dsdt2 := ( cos( t ) cos( 3 t ) − 12 sin( 3 t ) − 3 sin( 3 t ) sin( t ) )2

+ ( cos( t ) sin( 3 t ) + 12 cos( 3 t ) + 3 cos( 3 t ) sin( t ) )2 + sin( t )2

> dsdt2 := simplify(dsdt2);

dsdt2 := −9 cos( t )2 + 72 sin( t ) + 154

The speed is

> dsdt := sqrt(dsdt2);

dsdt :=
√

−9 cos( t )2 + 72 sin( t ) + 154

The length of the arc from t = 0 to t = 2π is

> Length := Int(s,t=0..2*Pi);

Length :=
∫ 2 π

0

√
−9 cos( t )2 + 72 sin( t ) + 154 dt

A ten digit approximation for this is given by

> Length := evalf(Length);

Length := 75.66806064

Figure 82: Curve in Space Figure 83: A Non-linear Helix

We now return our expressions R, V, and ds
dt to the general case to discuss acceleration.

> R := vector([f(t),g(t),h(t)]); V:= map(diff,R,t); dsdt := sqrt(dotprod(V,V));

R := [ f( t )g( t )h( t ) ]

V :=
[

∂

∂t
f( t )

∂

∂t
g( t )

∂

∂t
h( t )

]

dsdt :=
√(

∂

∂t
f( t )

)2

+
(

∂

∂t
g( t )

)2

+
(

∂

∂t
h( t )

)2

The unit tangent vector T is the normalized tangent vector:

> T := scalarmul(V,1/dsdt);

T :=
[

∂
∂t f( t )√

%1

∂
∂t g( t )√

%1

∂
∂t h( t )√

%1

]

%1 :=
(

∂

∂t
f( t )

)2

+
(

∂

∂t
g( t )

)2

+
(

∂

∂t
h( t )

)2

One important fact about T and its derivative is that they are orthogonal. To see this calculate the derivative.

> dTdt := map(diff,T,t);

dTdt :=
[
− 1

2

(
∂
∂t f( t )

)
%5

%23/2 + %4√
%2

− 1
2

(
∂
∂t g( t )

)
%5

%23/2 + %3√
%2

− 1
2

(
∂
∂t h( t )

)
%5

%23/2 + %1√
%2

]

%1 := ∂2

∂t2
h( t )

%2 :=
(

∂

∂t
f( t )

)2

+
(

∂

∂t
g( t )

)2

+
(

∂

∂t
h( t )

)2

%3 := ∂2

∂t2
g( t )

%4 := ∂2

∂t2
f( t )

%5 := 2

(
∂

∂t
f( t )

)
%4 + 2

(
∂

∂t
g( t )

)
%3 + 2

(
∂

∂t
h( t )

)
%1

Now take the dot product of T and its derivative:

> simplify(dotprod(T,dTdt));

0

Thus T is perpendicular to its derivative.
The acceleration vector is equal to the derivative of the velocity vector:

> A := map(diff,V,t);

A :=
[

∂2

∂t2
f( t )

∂2

∂t2
g( t )

∂2

∂t2
h( t )

]
The radius of curvature is given by

> r := s/sqrt(dotprod(dTdt,dTdt)):

The curvature is the reciprocal of the radius of curvature:

> kappa := 1/r:

Example: Find the tangential and normal components of acceleration and the curvature for the helix given by the
equations for t > 0 :

x = 3 cos( t2 ), y = 3 sin( t2 ), z = t2

Solution: The position vector is

> R := vector([3*cos(t^2),3*sin(t^2),t^2]);

R := [ 3 cos( t2 )3 sin( t2 ) t2 ]

We aren’t asked to plot this curve for this problem but we will anyway. See Figure 83

> plots[spacecurve](R,t=0..Pi);

The velocity and acceleration vectors are:

> V := map(diff,R,t); A := map(diff,V,t);

V := [−6 sin( t2 ) t 6 cos( t2 ) t 2 t ]

A := [−12 cos( t2 ) t2 − 6 sin( t2 ) − 12 sin( t2 ) t2 + 6 cos( t2 )2 ]

The speed and unit tangent vectors are

> dsdt := simplify(sqrt(simplify(dotprod(V,V))),symbolic);

dsdt := 2
√

10 t

> T := scalarmul(V,1/dsdt);

T :=
[
− 3

10

√
10 sin( t2 )

3
10

√
10 cos( t2 )

1
10

√
10

]
Using vector analysis we know that the tangential component of acceleration is

dotprod(A, T) · T.

> Atan := scalarmul(T,dotprod(A,T));

Atan :=
[
− 3

10
%1

√
10 sin( t2 )

3
10

%1
√

10 cos( t2 )
1

10
%1

√
10

]
%1 := − 3

10
(−12 cos( t2 ) t2 − 6 sin( t2 ) )

√
10 sin( t2 )

+ 3
10

(−12 sin( t2 ) t2 + 6 cos( t2 ) )
√

10 cos( t2 ) + 1
5

√
10

> simplify(%1);

2
√

10

> Atan := map(z->subs(%1=2*sqrt(10),z),Atan);

Atan := [−6 sin( t2 )6 cos( t2 )2 ]

> absAtan := sqrt(simplify(dotprod(Atan,Atan)));

absAtan := 2
√

10

Of course the other formula for this quantitiy is d2s
dt2 :

> diff(dsdt,t);

2
√

10

If you know the acceleration vector and its tangential component then by vector subtraction you can get the
normal component of acceleration.

> Anormal := evalm(A-Atan);

Anormal := [−12 cos( t2 ) t2 − 12 sin( t2 ) t2 0 ]

The magnitude of this component is thus

> simplify(sqrt(simplify(dotprod(Anormal,Anormal))),symbolic);

12 t2

This means that the curvature is calculated by

> kappa := "/dsdt^2;

κ := 3
10

Another way to obtain the normal component is from the formula:

ds
dt · dT/dt.

> dTdt := map(diff,T,t);

dTdt :=
[
− 3

5

√
10 cos( t2 ) t − 3

5

√
10 sin( t2 ) t 0

]
> scalarmul(dTdt,dsdt);

[−12 cos( t2 ) t2 − 12 sin( t2 ) t2 0 ]

This agrees with the previous expression fo the normal component of acceleration.

Exercises 14.3

1. If
R(t) = sin(2t)i + 3 cos(3t)j + t2k,

find V, A, ds
dt , and the approximate value of the arc length of the curve as t varies form 0 to 2π. Plot a the

curve over this interval.

14.4 Vector Fields

There are several graphing procedures in Maple V that help you to study vector fields.

> with(plots):

Within the plots package there is the command gradplot that allows you to plot the gradient vector field for
a function of two variables. For a plot of the gradient vector field of the scalar function f (x, y) = x2 + 2y2 see
Figure 84.

Figure 84: Gradient Plot of x2 + 2y2

Figure 85: Gradient Plot with some Level Curves

> Q := gradplot(x^2+2*y^2,x=-3..3,y=-3..3):";

This command along with implicitplot allows you to plot the vector field along with several level curves. See
Figure 85.

> P.1 := implicitplot(x^2+2*y^2=1,x=-3..3,y=-3..3):";
> for i from 1 to 3 do P.(2*i+1) := implicitplot(x^2+2*y^2=2*i+1,
> x=-3..3,y=-3..3):od:
> display({Q,seq(P.(2*i-1),i=1..4)});

If you wish to view the flow the vector field in two dimensions, then it is useful to utilize the tools that are part
of the DEtools package.

> with(DEtools):

We first use grad to calculate the gradient vector field of the function.

> f := x^2+2*y^2;

f := x2 + 2y2

The linalg package needs to be made available for some of the commands that follow.

> with(linalg):

Warning: new definition for norm
Warning: new definition for trace

> X := grad(f,[x,y]);

X := [ 2 x 4 y ]

However, to use this vector field as an argument for DEplot2 we must convert it to the type list.

> X := convert(X,list);

X := [ 2 x, 4 y ]

The next command plots the gradient vector field along with the three flowlines which start at the points (−1.5, 1.1), (1.1, 0.83),

and (0.8, 0.9). See Figure 86.

Figure 86: Gradient Vector Field and some Flow Lines

Figure 87: Gradient Vector Field with Level Curves
and Flow Lines

> Flow := DEplot2(X,[x,y],-0.5..0.5,
> {[0,-1.5,1.1],[0,1.1,0.83],[0,0.8,0.9]}):";

Now if we want some level curves for f (x, y), along with the vector field and flowlines, we do something like
the following. See Figure 87.

> P.1 := implicitplot(x^2+2*y^2=1,x=-4..3,y=0..9):";
> for i from 1 to 5 do P.(2*i) := implicitplot(x^2+2*y^2=(2*i)^2,
> x=-4..3,y=0..9):od:
> display({Flow,seq(P.(2*i),i=1..5)},scaling=constrained);

It is also possible to plot two and three dimensional vector fields using fieldplot and fieldplot3d. For example to
plot in Figure 88 of the vector field

< 2x, y, x − z >

execute the following commands:

> X := vector([2*x,y,x-z]);

X := [ 2 x y x − z ]

> fieldplot3d(X,x=-3..3,y=-3..3,z=-3..3);

Exercises 14.4

1. Plot the level curves
f (x, y) = x2 − xy + y2 = j, for j = 1, 2, 3

on a single Maple V plot with −4 ≤ x ≤ 4, −4 ≤ y ≤ 4. Make a plot of the gradient vector field along with
flow lines through the points (−1.5, 1), (1, 1), (1, 2), and (1,−3).

Figure 88: A Three Dimensional Field Plot

2. Plot flow lines for the vector field
F(x, y) =< −2y, x >

that go through points (1, 0), (2, 0), and (3, 0).

14.5 Line Integrals with Maple V

In this section we compute line integrals of vector fields over curves in two and three dimensional space.

> with(linalg):

For a start lets compute the line integral of the vector field < x2, x > over the straight line that connects (0,0)
to (1,0).

> V := vector([x^2,x]);

V := [ x2 x ]

> r := vector([t,0]);

r := [ t 0 ]

> dr := map(diff,r,t);

dr := [ 1 0 ]

> B := subs({x=t,y=0},dotprod(V,dr));

B := t2

> Int(V,r) = int(B,t=0..1); ∫
V dr = 1

3

As another example let’s integrate the vector field

X =< xy, yz, xz >

over the twisted cubic, which is given parametrically by

x = t, y = t2, z = t3,

from (−1, 1,−1) to (1, 1, 1).

> V := vector([x*y,y*z,x*z]); r := vector([t,t^2,t^3]);

V := [ x y y z x z ]

r := [ t t2 t3 ]

> dr := map(diff,r,t);

dr := [ 1 2 t 3 t2 ]

> B := subs({x=t,y=t^2,z=t^3},dotprod(V,dr));

B := t3 + 5 t6

> Int(V,r) = int(B,t=-1..1);

∫
V dr = 10

7

Exercises 14.5

1. Let F(x, y, z) =< x3 y, xy, yz > . Find the work done by F along a circle of radius 5 centered at the origin
in the yz plane.

2. Evaluate the line integral of the vector field F(x, y, z) =< xye−y, xz, sin(z) > along the parabola z =
x2, y = 0, between (−1, 0, 1) and (1, 1, 1).

14.6 Divergence and Curl in Maple V

We can compute the curl and divergence of a vector field by using commands that are contained in the linalg pack-
age.

> with(linalg):

Warning: new definition for norm
Warning: new definition for trace

As an example we will calculate the divergence and curl of the vector field

< sin x, exp xyz, x + sin z > .

> F := vector([sin(x),exp(x*y*z),x+sin(z)]);

F := [ sin( x ) e( x y z ) x + sin( z ) ]

> divF := diverge(F,[x,y,z]);

divF := cos( x ) + x z e( x y z ) + cos( z )

> curlF := curl(F,[x,y,z]);

curlF := [−x y e( x y z ) − 1 y z e( x y z ) ]

If we need to find the value of these quantities at a particular point, say (0, 1, π), then we can use subs and
eval.

> ValuedivF := subs(x=0,y=1,z=Pi,divF);

ValuedivF := cos( 0 ) + cos( π )

> ValuedivF := simplify(ValuedivF);

ValuedivF := 0

> ValuecurlF := subs(x=0,y=1,z=Pi,eval(curlF));

ValuecurlF := [ 0 − 1 π e0 ]

> ValuecurlF := map(simplify,ValuecurlF);

ValuecurlF := [ 0 − 1 π ]

Note that there was a little difference in finding these values. The divergence is a scalar expression and so a
simple use of simplify was able to simplify it. On the other hand the curl of a vector field is a vector and required
the use of eval and map to simplify it completely.

Exercises 14.6

1. Compute the divergence and curl for the vector field

F(x, y, z) =< x2 sin(yz), y2 − xz, 2x2zez2
).

Evaluate the divergence and curl at the point (1,2,-1).

14.7 Surface Integrals with Maple V

Maple V can be useful in calculating surface integrals of vector fields over a parameterized surface. Let

F(x, y, z) =< f (x, y, z), g(x, y, z), h(x, y, z) >

denote an arbitrary vector field in three dimensional space. We will define it to a Maple V session in the following.
We assume that a surface S is parameterized with

R(u, v) =< x(u, v), y(u, v), z(u, v) > .

> f := ’f’:g:= ’g’:h:=’h’:

> with(linalg):
Warning: new definition for norm
Warning: new definition for trace

> F := vector([f(x,y,z),g(x,y,z),h(x,y,z)]);

F := [ f( x, y, z )g( x, y, z )h( x, y, z ) ]

The surface S is defined by:

> R := vector([x(u,v),y(u,v),z(u,v)]);

R := [ x( u, v )y( u, v ) z( u, v ) ]

The tangent vectors to the level curves v = constant and u = constant are computed as follows:

> Ru := map(diff,R,u);

Ru :=
[

∂

∂u
x( u, v )

∂

∂u
y( u, v )

∂

∂u
z( u, v )

]
> Rv := map(diff,R,v);

Rv :=
[

∂

∂v
x( u, v )

∂

∂v
y( u, v )

∂

∂v
z( u, v )

]
The normal, NS, to the surface for the given paramterization is defined by the following.

> NS := crossprod(Ru,Rv);

N S :=
[(

∂

∂u
y( u, v )

) (
∂

∂v
z( u, v )

)
−

(
∂

∂u
z( u, v )

) (
∂

∂v
y( u, v )

)
(

∂

∂u
z( u, v )

) (
∂

∂v
x( u, v )

)
−

(
∂

∂u
x( u, v )

) (
∂

∂v
z( u, v )

)
(

∂

∂u
x( u, v )

) (
∂

∂v
y( u, v )

)
−

(
∂

∂u
y( u, v )

) (
∂

∂v
x( u, v )

)]
The flux integral ∫

S
F · NS d A

for the vector field can now be computed

> Int(F,R=S..NULL) = Int(Int(subs(x=x(u,v),y=y(u,v),z=z(u,v),
> dotprod(F,NS)),u=a..b),v=c..d);

∫
S

F d R =
∫ d

c

∫ b

a
f( x( u, v ), y( u, v ), z( u, v ) )

((
∂

∂u
y( u, v )( u, v )

) (
∂

∂v
z( u, v )( u, v )

)
−

(
∂

∂u
z( u, v )( u, v )

) (
∂

∂v
y( u, v )( u, v )

))
+ g( x( u, v ), y( u, v ), z( u, v ) )

(
(

∂

∂u
z( u, v )( u, v )

) (
∂

∂v
x( u, v )( u, v )

)
−

(
∂

∂u
x( u, v )( u, v )

) (
∂

∂v
z( u, v )( u, v )

))
+ h( x( u, v ), y( u, v ), z( u, v ) )

(
(

∂

∂u
x( u, v )( u, v )

) (
∂

∂v
y( u, v )( u, v )

)
−

(
∂

∂u
y( u, v )( u, v )

) (
∂

∂v
x( u, v )( u, v )

))
dudv

Example: Evaluate the surface integral of the vector field

F =< x, y,−2z >

over the upper half of the sphere

x2 + y2 + z2 = a2

Solution: One parameterization is given by

x = a cos( u ) sin( v ), y = a sin( u ) sin( v ), z = a sin( v )

> with(linalg):

Warning: new definition for norm Warning: new
definition for trace

> F := vector([x,y,-2*z]);

F := [ x y − 2 z ]

> R := vector([a*cos(u)*cos(v),a*sin(u)*cos(v),a*sin(v)]);

R := [ a cos( u ) cos( v ) a sin( u ) cos( v ) a sin( v ) ]

> Ru := map(diff,R,u);

Ru := [−a sin( u ) cos( v ) a cos( u ) cos( v )0 ]

> Rv := map(diff,R,v);

Rv := [−a cos( u ) sin( v ) − a sin( u ) sin( v ) a cos( v ) ]

> NS := crossprod(Ru,Rv);

N S := [a2 cos( u ) cos( v )2 a2 sin( u ) cos( v )2

a2 sin( u )2 cos( v ) sin( v ) + a2 cos( u )2 cos( v ) sin( v )]

> NS := map(simplify,NS);

N S := [ a2 cos( u ) cos( v )2 a2 sin( u ) cos( v )2 a2 cos( v ) sin( v ) ]

The integrand for the integral is

> H := subs(x=R[1],y=R[2],z=R[3],dotprod(NS,F));

H := a3 cos( u )2 cos( v )3 + a3 sin( u )2 cos( v )3 − 2 a3 cos( v ) sin( v )2

> H := simplify(H);

H := 3 a3 cos( v )3 − 2 a3 cos( v )

> int(int(H,u=0..Pi/2),v=0..2*Pi);

0

Thus the value of this surface integral is 0.
As another example consider the surface S which is bounded below by

z = x2 + y2

and above by the paraboloid

z = 4 − x2 − y2.

> with(plots):

We define our surface with the expressions

> z:= ’z’: y := ’y’: x := ’x’: bottom := x^2+y^2;

bottom := x2 + y2

> top := 4 -x^2-y^2;

top := 4 − x2 − y2

and plot the surface as shown in Figure 89

> B:= cylinderplot(sqrt(z),theta=0..2*Pi,z=0..2):";

> T := cylinderplot(sqrt(4-z),theta=0..2*Pi,z=2..4):";

> display3d({B,T},axes=NORMAL,style=PATCH,scaling=CONSTRAINED,

> orientation=[45,75]);

Figure 89: Figure Formed by Two Paraboloids

Let us assume that the density of the material that the surface is composed of is

d( x, y, z ) = z
π

+ 1
π

and our problem is to calculate the total mass of the surface. This amounts to computing the surface integral of
d(x, y, z) over each of the two parts: the top and the bottom and adding the results.

> d := z/Pi+1/Pi;

d := z
π

+ 1
π

Parameterizations for these surfaces are

> Rbottom := vector([u,v,u^2+v^2]);

Rbottom := [ u v u2 + v2 ]

> Rtop :=vector([u,v,4-u^2-v^2]);

Rtop := [ u v 4 − u2 − v2 ]

The normal for the bottom will now be computed.

> Rbottomu := map(diff,Rbottom,u); Rbottomv:= map(diff,Rbottom,v);

Rbottomu := [ 1 0 2 u ]

Rbottomv := [ 0 1 2 v ]

> Nbottom := crossprod(Rbottomu,Rbottomv);

Nbottom := [−2 u − 2 v 1 ]

For this problem below we want Nbottom to have z component pointing down so write:

> Nbottom := scalarmul(Nbottom,-1);

Nbottom := [ 2 u 2 v − 1 ]

> magNBottom := sqrt(dotprod(Nbottom,Nbottom));

magN Bottom :=
√

4 u2 + 4 v2 + 1

The integrand is thus

> H := subs(x=u,y=v,z=u^2+v^2,d)*magNBottom;

H :=
(

u2 + v2

π
+ 1

π

) √
4 u2 + 4 v2 + 1

It follows that a double integral which is equal to the mass of the bottom surface is given by the following:

> Bmass := Int(Int(H,v=-sqrt(2-u^2)..sqrt(2-u^2)),u=-sqrt(2)..sqrt(2));

Bmass :=
∫ √

2

−√
2

∫ √
2−u2

−
√

2−u2

(
u2 + v2

π
+ 1

π

) √
4 u2 + 4 v2 + 1 dv du

This integral is pretty hard to evaluate even with Maple V

> value(Bmass);

∫ √
2

−√
2

3
64

(24 u2
√

2 − u2 + 24 ln( 2
√

2 − u2 + 3 ) u2 + 66
√

2 − u2

+ 16 u4 ln( 2
√

2 − u2 + 3 ) + 5 ln( 2
√

2 − u2 + 3 ))
/
π − 3

64
(−24 u2

√
2 − u2

+ 24 ln(−2
√

2 − u2 + 3 ) u2 − 66
√

2 − u2 + 16 u4 ln(−2
√

2 − u2 + 3 )

+ 5 ln(−2
√

2 − u2 + 3 ))
/
πdu

Thus Maple V was unable to compute the exact value of this integral in its present form. But we can get a
numerical approximation using evalf.

> evalf(");

9.300000000

If we use our head a little bit we can simplify the preceding double integral with polar coordinates, then · · ·
> Bmass := Int(Int((r^2+1)/Pi*sqrt(4*r^2+1)*r,r=0..sqrt(2)),
> theta=0..2*Pi);

Bmass :=
∫ 2 π

0

∫ √
2

0

( r2 + 1 )
√

4 r2 + 1 r
π

dr dθ

> Bmass := value(");

Bmass := 93
10

Now we do the same thing for the top

> Rtopu := map(diff,Rtop,u); Rtopv := map(diff,Rtop,v);

Rtopu := [ 1 0 − 2 u ]

Rtopv := [ 0 1 − 2 v ]

> Ntop := crossprod(Rtopu,Rtopv);

Ntop := [ 2 u 2 v 1 ]

> magNtop := sqrt(dotprod(Ntop,Ntop));

magNtop :=
√

4 u2 + 4 v2 + 1

> H := subs(x=u,y=v,z=4-u^2-v^2,d)*magNtop;

H :=
(

4 − u2 − v2

π
+ 1

π

) √
4 u2 + 4 v2 + 1

Converting to polar coordinates we have

> H := subs(u=r*cos(theta),v=r*sin(theta),H);

H :=
(

4 − r2 cos( θ )2 − r2 sin( θ )2

π
+ 1

π

) √
4 r2 cos( θ )2 + 4 r2 sin( θ )2 + 1

> H := simplify(H);

H := −
√

4 r2 + 1 (−5 + r2 )

π

> Tmass := Int(Int(H*r,r=0..sqrt(2)),theta=0..2*Pi);

Tmass :=
∫ 2 π

0

∫ √
2

0
−

√
4 r2 + 1 (−5 + r2 ) r

π
dr dθ

> Tmass := value(Tmass);

Tmass := 167
10

> Total := Bmass + Tmass;

Total := 26

We conclude that the total mass of the surface is 26.
Let us continue with the surface that was defined in the last example and assume that we have a vector field

F =< x, y, 1 >

given. Now the surface integral of this vector field over the bottom surface is

> F := vector([x,y,1]);

F := [ x y 1 ]

We now make use of the fact that we oriented the normal vector for the bottom surface.

> H := subs(x=u,y=v,z=sqrt(u^2+v^2),dotprod(F,Nbottom));

H := 2 u2 + 2 v2 − 1

> Bflux :=
> Int(Int(H,v=-sqrt(2-u^2)..sqrt(2-u^2)),u=-sqrt(2)..sqrt(2));

Bf lux :=
∫ √

2

−√
2

∫ √
2−u2

−
√

2−u2
2 u2 + 2 v2 − 1 dv du

> Bflux := value(Bflux);

Bf lux := 2 π

Now we do the same thing using the top.

> H:= subs(x=u,y=v,z=sqrt(4-u^2-v^2),dotprod(F,Ntop));

H := 2 u2 + 2 v2 + 1

> Tflux := Int(Int(H,v=-sqrt(2-u^2)..sqrt(2-u^2)),u=-sqrt(2)..sqrt(2));

T f lux :=
∫ √

2

−√
2

∫ √
2−u2

−
√

2−u2
2 u2 + 2 v2 + 1 dv du

> Tflux := value(");

T f lux := 6 π

> SurfaceFlux := Tflux+Bflux;

Sur f aceFlux := 8 π

We conclude that the flux through the surface due to our vector field is 8π. Can we verify this value using the
Divergence Theorem? We start by computing the divergence of F.

> divF := diverge(F,[x,y,z]);

divF := 2

> SurfaceFlux := Int(Int(Int(divF,z=bottom..top),y=-sqrt(2-x^2)..

> sqrt(2-x^2)),x=-sqrt(2)..sqrt(2));

Sur f aceFlux :=
∫ √

2

−√
2

∫ √
2−x2

−
√

2−x2

∫ 4−x2−y2

x2+y2
2 dz dy dx

> SurfaceFlux := value(SurfaceFlux);

Sur f aceFlux := 8 π

This verifies Gauss’s Theorem for this problem since both methods for computing the surface flux yield the
same answer.

Example: Verify Stoke’s Theorem for

F =< 2x − y,−yz2,−y2z >,

where S is the upperhalf surface of the sphere

x2 + y2 + z2 = 1

and C is the boundary.

Solution: First we calculate the line integral of the vector field over C. The circle C is given by

> R := vector([cos(t),sin(t),0]);

R := [ cos( t ) sin( t )0 ]

We now prepare to find the value of the integral that occurs on the right hand side of Stoke’s Theorem.

> dR := map(diff,R,t);

d R := [−sin( t ) cos( t )0 ]

> F := vector([2*x-y,-y*z^2,-y^2*z]);

F := [ 2 x − y − y z2 − y2 z ]

> H := subs(x=cos(t),y=sin(t),z=0,dotprod(F,dR));

H := −( 2 cos( t ) − sin( t ) ) sin( t )

> RightSide := Int(H,t=0..2*Pi);

RightSide :=
∫ 2 π

0
− ( 2 cos( t ) − sin( t ) ) sin( t ) dt

> RightSide := value(RightSide);

RightSide := π

Now we calculate the value of the integral on the left hand side of the equality in Stoke’s Theorem. First we
obtain a parametrization for the upper hemisphere:

x = cos( u ) cos( v ), y = sin( u ) cos( v ), z = sin( v )

> R:= vector([cos(u)*cos(v), sin(u)*cos(v),sin(v)]);

R := [ cos( u ) cos( v ) sin( u ) cos( v ) sin( v ) ]

> Ru := map(diff,R,u);

Ru := [−sin( u ) cos( v ) cos( u ) cos( v )0 ]

> Rv := map(diff,R,v);

Rv := [−cos( u ) sin( v ) − sin( u ) sin( v ) cos( v ) ]

> NS := map(simplify,eval(crossprod(Ru,Rv)));

N S := [ cos( u ) cos( v )2 sin( u ) cos( v )2 cos( v ) sin( v ) ]

> curlF := curl(F,[x,y,z]);

curlF := [ 0 0 1 ]

> H:= subs(x=cos(u)*cos(v),y=sin(u)*cos(v),z=sin(v),dotprod(curlF,NS));

H := cos( v ) sin( v )

> LeftHand := Int(Int(H,u=0..2*Pi),v=0..Pi/2);

Le f tHand :=
∫ 1/2 π

0

∫ 2 π

0
cos( v ) sin( v ) du dv

> LeftHand:= value(LeftHand);

Le f tHand := π

Thus the two integrals have the same value as predicted by Stoke’s Theorem.

Exercises 14.7

1. Let F(x, y, z) =< 2xy, 3yz, −4z2 >, and S be the sphere x2 + y2 + z2 = 25, oriented so that the normal
points outward. Verify the Divergence Theorem in this case by evaluating the flux integral of the vector field
F over the surface, and then by computing the triple integral of div F over the interior of S.

2. Compute the line integral of the vector field F(x, y, z) =< x2, z − yx, xz > over the curve of intersection
of the sphere x2 + y2 + z2 = 1, with the plane y = z. Assume this curve is oriented so that its projection on
the xy plane is oriented clockwise. Calculate the integral directly and then use Stoke’s Theorem to check
your answer.

