
12 Differentiation of Functions of More Than One Variable

12.1 Partial Derivatives with Maple V

Let f be a function of two variables defined around the point (a, b). Recall that the definition of the partial deriva-
tive with respect to x at (a, b) is given by the following limit (provided that the limit exists).

∂ f
∂x

( a, b ) = lim
h→0

f( a + h, b ) − f( a, b )

h

Similarly, the partial derivative with respect to y at (a, b) is

∂ f
∂y

( a, b ) = lim
k→0

f( a, b + k ) − f( a, b )

k
.

As an example of how Maple V can be used to compute the partial derivatives of the function:

f( x, y ) = x4 y2 − x y + 7

at the point (x,y) we first define f and take the appropriate limits.

> f := (x,y) -> x^4*y^2-x*y+7;

f := ( x, y ) → x4 y2 − x y + 7

> Limit((f(x+h,y)-f(x,y))/h,h=0):"= value(");

lim
h→0

( x + h )4 y2 − ( x + h ) y − x4 y2 + x y
h

= y ( 4 y x3 − 1 )

> Limit((f(x,y+k)-f(x,y))/k,k=0):"= value(");

lim
k→0

x4 ( y + k )2 − x ( y + k ) − x4 y2 + x y
k

= x ( 2 y x3 − 1 )

The preceding Maple V segment tells us that

∂ f
∂x

( x, y ) = y ( 4 y x3 − 1 )

and
∂ f
∂y

( x, y ) = x ( 2 y x3 − 1 )

Normally we will simply ask Maple V to compute the partial derivatives directly. We can use the inert form
for diff if we want to format the output.

> Diff(f(x,y),x): "= value(");

∂

∂x
( x4 y2 − x y + 7 ) = 4 x3 y2 − y

We need to factor the preceding result in order to see that it agrees with the limit defined above as the partial
derivative of f with respect to x.

> factor(");

∂

∂x
( x4 y2 − x y + 7 ) = y ( 4 y x3 − 1 )

Now we will use diff to compute the partial with respect to y.

> diff(f(x,y),y);

2 x4 y − x

> factor(");

x ( 2 y x3 − 1 )

It is possible to compute higher order derivatives. For example, the second derivative with respect to x can be
obtained as follows:

> Diff(f(x,y),x,x): "=value(");

∂2

∂x2
( x4 y2 − x y + 7 ) = 12 x2 y2

or

> diff(f(x,y),x,x);

12 x2 y2

or

> diff(f(x,y),x$2);

12 x2 y2

In the last step we used the $ operator which sometimes shortens the notation.

> x$5;

x, x, x, x, x

> x$3,y$2;

x, x, x, y, y

To compute the fifth order derivative
∂5 f

∂x3∂y2

we write:

> diff(f(x,y),x$3,y$2);

48 x

or

> Diff(f(x,y),x$3,y$2): "= value(");

∂5

∂y2 ∂x3
( x4 y2 − x y + 7 ) = 48 x

It is easy to compute the value of partial derivatives at a particular point, say (3,2), using subs.

> diff(f(x,y),x$2,y);

24 x2 y

> subs(x=3,y=2,");

432

Exercises 12.1 Find the indicated partial derivatives.

1. ∂3

∂x2∂y (eπx2 y)

2. ∂8

∂x3∂y5

(
cos(x−y)

x4

)

12.2 Local Linearity

In the same way that the tangent line at a point can be computed for a function of one variable, one can compute
the tangent plane for a function of two variables. The tangent plane for a function z = f (x, y) at the point (a, b)

is given by
z = f (a, b) + fx(a, b)(x − a) + fy(a, b)(y − b).

The function given by the right-hand side of the last equation is called the linear approximation of f (x, y) at
(a, b). As an example let

g( x, y ) = 4 − x2 − y2

> g := (x,y) -> 4-x^2-y^2;

g := ( x, y ) → 4 − x2 − y2

Now compute the tangent plane to z = g(x, y) at (3,4).

> gx(3,2) := subs(x=3,y=2,diff(g(x,y),x));

gx( 3, 2 ) := −6

> gy(3,2) := subs(x=3,y=2,diff(g(x,y),y));

gy( 3, 2 ) := −4

> LinApprx := (x,y)-> g(3,2)+gx(3,2)*(x-3)+gy(3,2)*(y-2);

LinApprx := ( x, y ) → g( 3, 2 ) + gx( 3, 2 ) ( x − 3 ) + gy( 3, 2 ) ( y − 2 )

> LinApprx(x,y);

17 − 6 x − 4 y

In Figure 40 you can compare the graph of g(x, y) with its linearization. This figure is constructed by the
following Maple V segment.

> P1 := plot3d(g(x,y),x=-4..4,y=-4..4,style=patch):";

> P2 := plot3d(LinApprx(x,y),x=-2..4,y=-2..3):";

> plots[display3d]({P1,P2},axes=boxed,view=-20..30);

In order to get an idea as to how well the linear approximation does its job consider the error

|g(x, y) − LinApprx(x, y)|

in the approximation where (x, y) is taken from the sequence of points

(4, 3), (3.8, 2.8), (3.6, 2.6), (3.4, 2.4), (3.2, 2.2), and (3, 2).

Using the seq we can observe how the error decreases as this sequence of points goes to (3,2).

> seq(evalf(abs(g(4-i/5,3-i/5)-LinApprx(4-i/5,3-i/5))),i=0..5);

Figure 40: Paraboloid g( x, y ) = 4 − x2 − y2 and its Tangent Plane at (3,2).

2., 1.280000000, .7200000000, .3200000000, .08000000000, 0

The following command illustrates how the zip command can be used to obtain the same error terms.

> zip((x,y) -> abs(g(x,y)-LinApprx(x,y)),

> [4,3.8,3.6,3.4,3.2,3],[3,2.8,2.6,2.4,2.2,2]);

[ 2, 1.28, .72, .32, .08, 0 ]

As one might expect the error gets smaller as (x,y) approaches (3,2).
Recall that another way to find linear approximations to a function of one variable is to use Taylor Polyno-

mials. The next Maple V segment illustrates how the command mtaylor can be used to obtain the same linear
approximation that was found above.

> readlib(mtaylor);

proc() ... end

> mtaylor(g(x,y),[x=3,y=2],2);

17 − 6 x − 4 y

Thus the multivariable Taylor Polynomial of degree 1 is the same as the linear approximation.

Exercises 12.2 Find the linear approximation of

f (x, y) = sin(x2 + y2)

near the point (
√

π
2 ,

√
π
2 ). Make a Maple V plot showing the graph of f (x, y) and its linear approximation.

12.3 Gradients and Directional Derivatives with Maple V

Let f be a function which has partial derivatives around the point (a, b). Let U =< u1, u2 > be a unit vector. Then
the directional derivative is defined to be the following limit (assuming that it exists):

Df( a, b ) = lim
h→0

f( a + h u1, b + h u2 ) − f( a, b )

h

Thus the directional derivative represents the instantaneous rate of change of f as (x, y) moves in the direction
U =< u1, u2 > from the point (a, b).

As example find the directional derivative of the function

g( x, y ) = x2 + y2 − x cos( π y ) − y sin( π x )

in the direction U =< 2/
√

5, 1/
√

5 > at the point (−1, 2).

First define the function g(x, y).

> g := (x,y) -> x^2+y^2-x*cos(Pi*y)-y*sin(Pi*x);

g := ( x, y ) → x2 + y2 − x cos( π y ) − y sin( π x )

Now apply the definition

> Dg(-1,2) := limit((g(-1+h*2/sqrt(5),2+h*1/sqrt(5))-g(-1,2))/h,h=0);

Dg(−1, 2 ) := − 2
5

√
5 + 4

5
π

√
5

That preceding Maple V output gives the rate of change of the values g(x, y) as (x, y) moves in the direction
of < 2/

√
5, 1/

√
5 > from the point (−1, 2). There is a more direct way to compute a directional derivative using

gradient vectors and vector operations. The function grad, gradient is part of the linalg package.

> with(linalg):

Warning: new definition for norm

Warning: new definition for trace

We now compute the gradient of g at an arbitrary point:

> Gradg := grad(g(x,y),[x,y]);

Gradg := [ 2 x − cos( π y ) − y cos( π x ) π 2 y + x sin( π y ) π − sin( π x ) ]

We use subs to evaluate the variables at the point (−1, 2).

> Gradg := subs(x=-1,y=2,eval(Gradg));

Gradg := [−2 − cos( 2 π ) − 2 cos(−π )π 4 − sin( 2 π )π − sin(−π ) ]

If we want to evaluate the cosine function at its arguments then eval must be used again along with map.

> Gradg := map(s->eval(s),Gradg);

Gradg := [−3 + 2 π 4 ]

We know that the directional derivative of a function f (x, y) at a point (a, b) in the direction U =< u1, u2 >is
also given by the dot product of the gradient of f at (a, b) and the vector U. Hence:

> Dg(-1,2) := dotprod(Gradg,[2/sqrt(5),1/sqrt(5)]);

Dg(−1, 2 ) := 2
5

(−3 + 2 π )
√

5 + 4
5

√
5

> Dg(-1,2) := simplify(Dg(-1,2));

Dg(−1, 2 ) := − 2
5

√
5 + 4

5
π

√
5

If you wish a decimal approximation of this then perform the following.

> evalf(Dg(-1,2));

4.725424596

Recall that the gradient vector is perpendicular to the level curves. In order to illustrate this, let us draw the
gradient vector field of g(x,y) in the plane. This can be done with the command gradplot which is part of the plots
package.

> with(plots):

> P1 := gradplot(g(x,y),x=-2..0,y=1..3,arrows=SLIM,grid=[30,30]):";

Let’s plot the level curve g(x, y) = g(−1, 2).

P2 := implicitplot(g(x,y)=g(-1,2),x=-2..0,y=1..3):";

Now putting both plots together we obtain. See Figure 41

display({P1,P2},scaling=constrained);

1

1.5

2

2.5

3

y

-2 -1.5 -1 -0.5 0
x

Figure 41: A Level Curve with Gradient Field

Figure 42: Level Curves with Gradient Field

Note that any place where an arrow from the gradient vector field intersects the level curve the vector is per-
pendicular to the curve. We will observe this with a few more level curves. See Figure 42

> P3:= implicitplot({g(x,y)=g(-0.5,2),g(x,y)=g(-1.5,1.5)},x=-2..0,y=1..3):";

> display({P1,P2,P3},scaling=constrained);

Exercises 12.3

1. If f (x, y, z) = 2xz2ez2 − x2 sin y2, find the gradient of f and its magnitude at (1, 1,
√

π/2).

2. If f (x, y, z) = 2xz2ez2 − x2 sin y2, find the directional derivative of f at (1, 1,
√

π/2) in the direction parallel
to the line joining (1, 2, 3) to (−3, 3, 6).



           

12 DIFFERENTIATION OF FUNCTIONS OF MORE THAN ONE VARIABLE 48

12.4 Quadratic Approximations

Just as one can use the Maple V procedure mtaylor to linearly approximate functions of more than one variable,
one can also find higher order approximations of these same functions. Using mtaylor it is easy to obtain Taylor
Polynomials of degree 2. The following Maple V segment shows how one can find the quadratic approximation
for the function

f (x, y) = sin x cos y

near the point (0, 0). First we make the procedure mtaylor available with a call to readlib.

> readlib(mtaylor);
proc() ... end

Now define the function f.

> f := (x,y)-> sin(x)*sin(y);

f := (x, y) 7→ sin(x) sin(y)

Now invoke the procedure and assign the approximation to the variable f 2.

> f2 := mtaylor(f(x,y),[x=0,y=0],3);

f 2 := xy

Next convert the expression f 2 to a function with the procedure unapply.

> f2 := unapply(f2,x,y);

f 2 := (x, y) 7→ xy

Some graphical comparisons of the function f with its quadratic approximation f 2 will now be made. Plots of f
and f 2 are created by the following two Maple V commands and shown in Figures 43 and 44 respectively.

> P1 := plot3d(f(x,y),x=-Pi..Pi,y=-Pi..Pi,orientation=[45,60],
> style=PATCHCONTOUR):";
> P2 := plot3d(f2(x,y),x=-Pi..Pi,y=-Pi..Pi,orientation=[45,60],
> style=PATCHCONTOUR ):";

In order to illustrate that the quadratic approximation is a pretty good estimate for f (x, y)

Figure 43: Plot of sin x cos y Figure 44: Plot of Quadratic Approximation of
sin x cos y

near the origin the Maple V plots[display] is used to display both functions simultaneously in Figure 45 over the
small square [−0.5, 0.5] × [−0.5, 0.5].

> plots[display]({P1,P2},view=[-.5..0.5,-.5..0.5,-1..1]);

The same functions are plotted over the larger square [−π,π] × [−π,π]. See Figure 46.

> plots[display]({P1,P2});

Figure 45: Plot of sin x cos y and its Quadratic Approx-
imation in a small neighborhood of (0, 0)

Figure 46: Plot of sin x cos y and its Quadratic Approx-
imation in a larger neighborhood of (0, 0)

Plotting contours of these two functions illustrate how the approximation is pretty good near the origin, but that
global features of the function of the function f are lost further away from the origin. See Figure 47 for a contour
plot of the function f. Compare this plot with Figure 48 which shows a contour plot of the quadratic approximation
f 2.

> plot3d(f(x,y),x=-Pi..Pi,y=-Pi..Pi,orientation=[270,0],
> style=contour);
> plot3d(f2(x,y),x=-Pi..Pi,y=-Pi..Pi,orientation=[270,0],
> style=contour);

Figure 47: Contour Plot of sin x cos y in a neighbor-
hood of (0, 0)

Figure 48: Contour Plot of Quadratic Approximation
in a neighborhood of (0, 0)

Exercises 12.4 Use mtaylor to find linear and quadratic approximations near the origin for each of the following
functions. Make graphical comparisons of each.

1. f (x, y) = e3x2
cos y

2. f (x, y) = sin(x − y)

3. f (x, y) = 1
1−x+y

12.5 Partial Differential Equations

Because of its ability in helping to find partial derivatives, Maple V can be used to verify when a function is a
solution of a partial differential. In addition, using graphics one can study qualitative features of the solutions. As
an example, consider the heat equation

∂

∂t
u = ∂2

∂x2
u

Using Maple V it easy to show that the function

u( x, t ) = −e(−4 t ) sin( 2 x )

is a solution.

> u := (x,t) -> -exp(-4*t)*sin(2*x);

u := ( x, t ) → −e(−4 t ) sin( 2 x )

Now differentiating we see that u satisfies the heat equation.

> LeftHandSide := diff(u(x,t),t);

LeftHandSide := 4 e(−4 t ) sin( 2 x )

> RightHandSide := diff(u(x,t),x,x);

RightHandSide := 4 e(−4 t ) sin( 2 x )

Suppose, for example, that u(x, t) represents the heat in a bar which is π units long. Initially the heat at point
x along the bar is u(x, 0).

> u(x,0);

−sin( 2 x )

We can plot this to obtain a graph of the temperature distribution along the bar at times t = 0 and t = 1/8,

respectively. See Figures 49 and 50.

> P.1 := plot(u(x,0),x=0..Pi):";

> P.2 := plot(u(x,1/8),x=0..Pi):";

Its rather difficult to see how one plot differs from the other, but if we place both plots on the same graph we get
an idea about how the temperature changes as t varies from 0 to 1/8. See Figure 51

> with(plots):
> display([P.1,P.2]);

In order to view changes as t changes from 2/8 to 7/8 in increments of 1/8 we perform the following do loop.

> for i from 3 to 8 do

> P.i := plot(u(x,(i-1)/8),x=0..Pi):

> od:

Now we can get a multiple plot that shows how the heat varies as t changes from 0 to 7/8 in increments of 1/8.
See Figure 52

> display([seq(P.i,i=1..8)]);



             

12 DIFFERENTIATION OF FUNCTIONS OF MORE THAN ONE VARIABLE 52

-1

-0.5

0

0.5

1

0 0.5 1 1.5 2 2.5 3
x

Figure 49: Graph of u(x, 0)

Figure 50: Graph of u(x, 1/8)

Figure 51: Graph of u(x, 0) and u(x, 1/8)

Figure 52: Multiplot of of u(x, i/8), i = 0 · · ·7

This same information can be animated by using the option insequence = true. Thus we obtain an animated
plot showing how the temperature in the bar varies as t varies from 0 to 7/8 in increments of 1/8.

> display([seq(P.i,i=1..8)],insequence=true);

Exercises 12.5

1. A twice continously differentiable function u(x, y) which satisfies the Laplace’s Differential Equation

∂2u
∂x2

+ ∂2u
∂y2

= 0

is said to be a harmonic function. Determine which of the following functions are harmonic.

(a) u(x, y) = eπx sin(πy)

(b) u(x, y) = x3 + y3

(c) u(x, y) = ln(x2 + y2)

(d) u(x, y) = arctan(y/x)

2. Verify that

y(x, t) = 3
40

cos(2t) sin(x) − 1
40

cos(6t) sin(3x)

satisfies the wave equation
∂2 y
∂t2

= 4
∂2 y
∂x2

, (0 < x < π, t > 0, )

and the boundary conditions

y(0, t) = y(π, t) = 0,

y(x, 0) = 3
40

sin(x) − 1
40

sin(3x).

12.6 Local and Global Extrema

In this section the problem of finding extreme values of functions of more than one variable will be studied. Con-
sider the function f defined as:

f( x, y ) = x y e(−1/2 x2−1/2 y2 )

The function can be defined into a Maple V session by the following command.

> f := (x,y) -> x*y*exp(-(x^2+y^2)/2);

f := ( x, y ) → x y e(−1/2 x2−1/2 y2 )

To get an idea about extreme values consider what a graph of this function shown in Figure 53 looks like.

Figure 53: Graph of xye−(x2+y2)/2 Figure 54: Contourplot of xye−(x2+y2)/2

> plot3d(f(x,y),x=-3..3,y=-3..3,style =patch,orientation=[70,65],
> axes=FRAMED);

This plot suggests that there appears to be an extreme value in each quadrant. There appears to be minima in
the second and fourth quadrants and maxima in the first and third quadrants.

The contour plot shown in Figure 54 is useful in finding the approximate location of critical points.

> with(plots):
> contourplot(f(x,y),x=-3..3,y=-3..3);

In order to find the exact location of the critical points for the function f analytically we need to determine the
first derivatives.

> fx := diff(f(x,y),x);

f x := y e(−1/2 x2−1/2 y2 ) − x2 y e(−1/2 x2−1/2 y2 )

The expression fx can be simplified.

> fx := factor(fx);

f x := −y e(−1/2 x2−1/2 y2 ) ( x − 1 ) ( x + 1 )

> fy := factor(diff(f(x,y),y));

f y := −x e(−1/2 x2−1/2 y2 ) ( y − 1 ) ( y + 1 )

The critical points are the points for which fx and fy vanish simultaneously.

> SOL := solve({fx=0,fy=0},{x,y});

SOL := { x = 0, y = 0 }, { x = 1, y = 1 }, { x = 1, y = −1 }, { x = −1, y = 1 },
{ x = −1, y = −1 }

Thus there are five critical points:

(0, 0), (1, 1), (1,−1), (−1, 1), and (−1,−1).

Inspection of the plots made above suggest that the points (1, 1) and (−1,−1) are maxima, and points (−1, 1), (1,−1)

are minima. Finally, the plots also suggest that (0, 0) is a saddle. In order to get a better focus on this plot the sur-
face in a neighborhood of each critical point.

First check out a neighborhood of (0, 0). See Figure 55.

Figure 55: Graph of xye−(x2+y2)/2 near (0, 0) Figure 56: Graph of xye−(x2+y2)/2 near (1, 1)

> plot3d(f(x,y),x=-0.1..0.1,y=-0.1..0.1,orientation = [70,65],
> style=patchcontour);

Indeed this Figure 55 provides further evidence that the point (0, 0) is a saddle. Now check the neighborhood
of the point (1, 1) which the previous plots suggest to be a maximum. See Figure 56.

> plot3d(f(x,y),x=0.9..1.1,y=0.9..1.1,orientation = [70,65]);

This point looks like a maximum. Finally, a plot around (-1,1) should also look like a minimum.

> plot3d(f(x,y),x=-1.1..-0.9,y=0.9..1.1,orientation = [70,65],
> style=patchcontour);

In the same way, one can obtain geometric evidence by plotting around the other two critical points, but instead
we will use analytic means to characterize the critical points.

Recall that if (a, b) is a critical point of f and

1 = fxx(a, b) fyy(a, b) − fxy(a, b)2,

then

(a) If 1 > 0 and fxx(a, b) > 0, (a, b) is a minimum.

(b) If 1 > 0 and fxx(a, b) < 0, (a, b) is a maximum.

(c) If 1 < 0, then (a, b) is a saddle point.

(d) If 1 = 0, anything can happen.

> fxx := factor(diff(f(x,y),x,x));

f xx := x y e(−1/2 x2−1/2 y2 ) (−3 + x2 )

> fxy := factor(diff(f(x,y),x,y));

f xy := e(−1/2 x2−1/2 y2 ) ( x − 1 ) ( x + 1 ) ( y − 1 ) ( y + 1 )

> fyy := factor(diff(f(x,y),y,y));

f yy := x y e(−1/2 x2−1/2 y2 ) (−3 + y2 )

> Delta := factor(fxx*fyy-fxy^2);

1 := −( e(−1/2 x2−1/2 y2 ) )2 (−5 x2 y2 + x2 y4 + x4 y2 + 1 − 2 x2 − 2 y2 + x4 + y4 )

We now apply the test using the critical point (0, 0) :

> eval(subs(x=0,y=0,Delta));

−1

Thus 1 < 0 at the point (0, 0) and this point is a saddle point.
Now check the values of fxx and 1 at the point (1, 1).

> subs(x=1,y=1,[fxx,Delta]);

[−2 e(−1 ), 4 ( e(−1 ) )2 ]

Since the number e is positive the last Maple V output means that fxx < 0, and 1 > 0 at (1, 1), which proves
that (1, 1) is the point where the maximum occurs. We can use evalf to approximate the values.

> evalf(");

[−.7357588824, .5413411332 ]

Next a check of the critical point (−1, 1) is made.

> subs(x=-1,y=1,[fxx,Delta]);

[ 2 e(−1 ), 4 ( e(−1 ) )2 ]

Thus (−1, 1) is a minimum.
Finally, for (−1,−1) and (1,−1).

> subs(x=-1,y=-1,[fxx,Delta]);subs(x=1,y=-1,[fxx,Delta]);

[−2 e(−1 ), 4 ( e(−1 ) )2 ]

[ 2 e(−1 ), 4 ( e(−1 ) )2 ]

You may conclude that (−1,−1) is a maximum and (1,−1) is a minimum.
The following Maple V segment shows how to obtain the value of f (x, y) at the four extreme values using

zip:

> zip(f,[1,-1,-1,1],[1,1,-1,-1]);

[ e(−1 ),−e(−1 ), e(−1 ),−e(−1 ) ]

The approximate extreme values to three decimal places are as follows:

> evalf(",3);

[ .368,−.368, .368,−.368 ]

The following table summarizes the results for this problem.

(a,b) fxx(a, b) 1 Conclusion f(x,y)
(0,0) 0 Negative Saddle Point 0
(1,1) Negative Positive Maximum .368
(-1,1) Positive Positive Minimum -.368
(-1,-1) Negative Positive Maximum .368
(1,-1) Positive Positive Minimum -.368

Exercises 12.6 Find an classify all critical points for the following functions of two variables. If an extreme value
is absolute then state why it is.

1. f (x, y) = xy − x2 − y2 − 2x − 2y + 4

2. f (x, y) = x3/3 + 9y3 − 4xy

3. f (x, y) = e−2x cos y

4. f (x, y) = xy + 2x − ln x2 y, x > 0, y > 0

12.7 Constrained Optimization

We illustrate how Maple V can be used to solve optimization problems that have constraints with the following
example.

Example: Find the dimensions of the box of largest volume which can be fitted inside the ellipsoid

x2

a2
+ y2

b2
+ z2

c2
= 1

assuming that each edge of the box is parallel to a co-ordinate axis.

Solution: Since each of the eight vertices of the box must lie on the ellipsoid, let the vertex in the first octant have
coordinates (x, y, z). The dimensions of the box are 2x, 2y, 2z, and its volume is V = 8xyz. The problem is to
find the values of (x, y, z), that lie on the ellipsoid and that make V as large as possible.

Thus we wish to optimize V = 8xyz subject to the constraint:

g( x, y, z ) = 1

where

g( x, y, z ) = x2

a2
+ y2

b2
+ z2

c2
.

Thus we must solve the equations given by

grad(V ) = λgrad(g),

for x, y, z, and λ, subject to the above constraint. First define V and g using Maple V.

> V := 8*x*y*z;

V := 8 x y z

> g := x^2/a^2+y^2/b^2+z^2/c^2;

g := x2

a2
+ y2

b2
+ z2

c2

> with(linalg):

Warning: new definition for norm
Warning: new definition for trace

The Maple V procedure grad from the linear algebra package can be used to compute the necessary gradients.

> GradV := grad(V,[x,y,z]);

GradV := [ 8 y z 8 x z 8 x y ]

> Gradg := grad(g,[x,y,z]);

Gradg :=
[
2

x
a2

2
y

b2
2

z
c2

]
Now one can solve the relevent equations for (x, y, z) and λ.

> SOL := solve({g=1,seq(GradV[i]=lambda*Gradg[i],i=1..3)},{x,y,z,lambda});

SOL := { x = 0, z = 0, λ = 0, y = b }, { x = 0, z = 0, λ = 0, y = −b },{
z = %1, y = %1 b

c
, x = − %1 a

c
, λ = −4 %1 a b

}
,{

z = %1, λ = −4 %1 a b, y = − %1 b
c

, x = %1 a
c

}
,{

z = %1, y = %1 b
c

, x = %1 a
c

, λ = 4 %1 a b

}
,{

z = %1, x = − %1 a
c

, y = − %1 b
c

, λ = 4 %1 a b

}
,

{ z = 0, y = 0, λ = 0, x = a }, { z = 0, y = 0, λ = 0, x = −a },
{ x = 0, y = 0, λ = 0, z = c }, { x = 0, y = 0, λ = 0, z = −c }
%1 := RootOf( 3 Z2 − c2 )

Warning: Maple V may return its output in this case in a variety of different but equivalent ways. For instance,
if you have typed in the very same commands as was done above, you may get different looking answers. In this
particular segment, Maple has returned ten sets of answers. We must now choose those that are correct. Since we
are seeking values of x, y, and z that are positive we can exclude all answers that cause these values to be otherwise.
First note that the symbol %1 is defined in terms of RootOf, a quadratic expression.

> allvalues(%1);

1
3

√
3 c,− 1

3

√
3 c

If we select the positive roots in each case and inspect the various ten sets of points in SOL above we see that
only one answer is eligible to be correct:

z = %1, y = %1b/c, x = %1a/c, λ = 4%1ab.

Remember that you may get a slightly different set of symbols here, but the results should still amount to the
same. Thus the desired values of xopt, yopt, and zopt are

> xopt:= allvalues(%1)[1]*a/c; yopt := allvalues(%1)[1]*b/c;

xopt := 1
3

√
3 a

yopt := 1
3

√
3 b

> zopt :=allvalues(%1)[1];

zopt := 1
3

√
3 c

With these values assigned we conclude that the maximum volume is

> subs(x=xopt,y=yopt,z=zopt,V);

8
9

a b
√

3 c

And with these values substituted into g we see that indeed (x, y, z) satisfy the constraint.

> subs(x=xopt,y=yopt,z=zopt,g);

1

Exercises 12.7

1. Find the maximum of the function z = −4x3 y2 where (x, y) must be a point on the unit circle centered at
the origin.

2. Given axbycz = A, find the maximum value of

(x + 1)(y + 1)(z + 1).

3. Find the maximum of
xm ynzp

subject to the constraint
x + y + z = a.

4. The Cobb-Douglas production model for a manufacturing process depending three inputs x, y, and z with
unit costs a, b, and c, respectively is given by

P = kxα yβzγ , α > 0, β > 0, γ > 0, α + β + γ = 1

subject to the constraint
ax + by + cz = d.

Determine x, y, and z to maximize P.

